Chapter 18

An Implementation of Single Class, Exact MVA

18.1. Introduction

In this appendix we provide a Fortran implementation of the most basic queueing network evaluation technique: the use of mean value analysis to obtain the exact solution of a separable queueing network model consisting entirely of queueing centers and containing a single class of batch type.

The algorithm on which this program is based is described in Chapter 6. The interested reader will find it educational to extend the program to accommodate two other characteristics described in that chapter: delay centers, and choice of batch, terminal, or transaction class types. The extension to multiple classes is given in Chapter 19. The extension to load dependent service centers is discussed in Chapter 20.

As noted in the overview of Part VI, our intention is that this program be used for educational experimentation with simple models. Its value as a capacity planning tool in no way approaches that of commercial queueing network modelling software. For a better idea of the interactions possible with that type of software, consult Chapter 16.

18.2. The Program

The program appears on the next two pages. Two statement labels (2001 and 2003) are included for reference in Chapter 20 and are not used in the program.

Note that some Fortran implementations impose restrictions on formatted I/O. It is best to include an explicit decimal point in real-valued input (but not integer-valued input) when using the program.

```
          program single
c
c A maximum of 25 centers are allowed.
c
          integer Ncusts,Ncents,n,center
          real demand(25)
          real qlen(25)
          real rtime(25)
          real tput,sysr
c
          write (6,5)
5         format (27h Input number of customers:)
          read (5,10) Ncusts
10        format (i4)
          write (6,15)
15        format (25h Input number of centers:)
          read (5,10) Ncents
          write (6,20)
20        format (25h Input service demand for)
          do 25 center=1,Ncents
             write (6,30) center
30           format (10h   Center ,i2,1h:)
             read (5,35) demand(center)
35           format (f8.4)
25           continue
c
c Now that the network is described, we perform the evaluation.
c Begin by initializing to the trivial solution for zero customers.
c
          do 40 center=1,Ncents
             qlen(center) = 0.0
40           continue
c
c The algorithm solves successively for each population.
c
          do 45 n=1,Ncusts
c
c First, compute the residence time at each center.
c
             sysr = 0.0
             do 50 center=1,Ncents
2001            rtime(center) = demand(center)*(1.0+qlen(center))
                sysr = sysr + rtime(center)
50              continue
```

```
c
c Next, use Little's law to compute system throughput.
c
                tput = n / sysr
c
c Finally, use Little's law to compute center queue lengths.
c
                do 55 center=1,Ncents
2003                qlen(center) = rtime(center) * tput
55                  continue
c
45              continue
c
c Print results.
c
            write (6,60) tput
60          format (20h System throughput: ,f8.4)
            write (6,65) Ncusts/tput
65          format (23h System response time: ,f8.4)
c
            write (6,70)
70          format (22h Device utilizations: )
            do 75 center=1,Ncents
                write (6,80) center,tput*demand(center)
80              format (i5,2h: ,f5.3)
75              continue
c
            write (6,85)
85          format (23h Device queue lengths: )
            do 90 center=1,Ncents
                write (6,95) center,qlen(center)
95              format (i5,2h: ,f8.4)
90              continue
c
            end
```